
F08 – Least-squares and Eigenvalue Problems (ScaLAPACK)

F08JJFP (PDSTEBZ)

NAG Parallel Library Routine Document

Note: before using this routine, please read the Users’ Note for your implementation to check for implementation-dependent
details. You are advised to enclose any calls to NAG Parallel Library routines between calls to Z01AAFP and Z01ABFP.

1 Description

F08JJFP (PDSTEBZ) computes eigenvalues of a real symmetric tridiagonal matrix T in parallel by
bisection. There are options to request all eigenvalues, all eigenvalues in the interval (vl, vu], or the
eigenvalues indexed il through iu.

Each logical processor must contain the vectors d and e which are the diagonal and off-diagonal elements
of T , respectively. A static partitioning of the workload is performed at the beginning of this routine
which results in all processes finding an (approximately) equal number of eigenvalues.

2 Specification

SUBROUTINE F08JJFP(ICNTXT, RANGE, ORDER, N, VL, VU, IL, IU,
1 ABSTOL, D, E, M, NSPLIT, W, IBLOCK, ISPLIT,
2 WORK, LWORK, IWORK, LIWORK, INFO)
ENTRY PDSTEBZ(ICNTXT, RANGE, ORDER, N, VL, VU, IL, IU,
1 ABSTOL, D, E, M, NSPLIT, W, IBLOCK, ISPLIT,
2 WORK, LWORK, IWORK, LIWORK, INFO)
DOUBLE PRECISION VL, VU, ABSTOL, D(∗), E(*), W(*), WORK(*)
INTEGER ICNTXT, N, IL, IU, M, NSPLIT, IBLOCK(*),
1 ISPLIT(*), LWORK, IWORK(*), LIWORK, INFO
CHARACTER∗1 RANGE, ORDER

The ENTRY statement enables the routine to be called by its ScaLAPACK name.

3 Usage
3.1 Definitions

None.

3.2 Global and Local Arguments

The following global input arguments must have the same value on entry to the routine on each processor
and the global output arguments will have the same value on exit from the routine on each processor:

Global input arguments: RANGE, ORDER, N, VL, VU, IL, IU, ABSTOL, D and E

Global output arguments: M, NSPLIT, W, IBLOCK, ISPLIT and INFO

The remaining arguments are local.

3.3 Distribution Strategy

Each logical processor (defined by the Library context ICNTXT) must contain the vectors d and e which
are the diagonal and off-diagonal entries of T , respectively.

3.4 Related Routines

The Library provides many support routines for the generation/distribution and input/output of data.
The following routines may be used in conjunction with F08JJFP (PDSTEBZ):

Real vector gather: F01ZPFP

4 Arguments

1: ICNTXT — INTEGER Local Input

On entry: the Library context, usually returned by a call to the Library Grid initialisation routine
Z01AAFP.

Note: the value of ICNTXT must not be changed.

2: RANGE — CHARACTER*1 Global Input

On entry: indicates which eigenvalues are required as follows:

if RANGE = ’A’, then all the eigenvalues are required;
if RANGE = ’V’, then all the eigenvalues in the half-open interval (vl, vu] are required;
if RANGE = ’I’, then eigenvalues with indices il to iu are required.

Constraint: RANGE = ’A’, ’V’ or ’I’.

3: ORDER — CHARACTER*1 Global Input

On entry: indicates the order in which the eigenvalues and their block numbers are to be stored as
follows:

if ORDER = ’B’, then the eigenvalues are to be grouped by split-off block and ordered from
smallest to largest within each block;
if ORDER = ’E’, then the eigenvalues for the entire matrix are to be ordered from smallest to
largest.

Constraint: ORDER = ’B’ or ’E’.

4: N — INTEGER Global Input

On entry: n, the order of the matrix T .

Constraint: N ≥ 0.

5: VL — DOUBLE PRECISION Global Input
6: VU — DOUBLE PRECISION Global Input

On entry: if RANGE = ’V’, the lower and the upper bounds, respectively, of the half-open interval
(vl, vu] within which the required eigenvalues lie.

Not referenced if RANGE = ’A’ or ’I’.

Constraint: VL < VU if RANGE = ’V’.

7: IL — INTEGER Global Input
8: IU — INTEGER Global Input

On entry: if RANGE = ’I’, il and iu, the indices of the first and the last eigenvalues, respectively,
to be computed (assuming that the eigenvalues are in non-decreasing order).

Not referenced if RANGE = ’A’ or ’V’.

Constraint: 1 ≤ IL ≤ IU ≤ N if RANGE = ’I’.

9: ABSTOL — DOUBLE PRECISION Global Input

On entry: the absolute tolerance to which each eigenvalue is required. An eigenvalue (or cluster) is
considered to have converged if it lies in an interval of width ≤ ABSTOL. If ABSTOL ≤ 0.0, then
the tolerance is taken as machine precision ×‖T ‖1.

10: D(∗) — DOUBLE PRECISION array Global Input

Note: the dimension of the array D must be at least max(1,N).

On entry: d, the diagonal elements of the tridiagonal matrix T .

11: E(∗) — DOUBLE PRECISION array Global Input

Note: the dimension of the array E must be at least max(1,N − 1).

On entry: e, the off-diagonal elements of the tridiagonal matrix T .

12: M — INTEGER Global Output

On exit: the actual number of eigenvalues found.

13: NSPLIT — INTEGER Global Output

On exit: the number of diagonal blocks which constitute the tridiagonal matrix T .

14: W(∗) — DOUBLE PRECISION array Global Output

Note: the dimension of the array W must be at least max(1,N).

On exit: the required eigenvalues of the tridiagonal matrix T stored in W(1) to W(M).

15: IBLOCK(∗) — INTEGER array Global Output

Note: the dimension of the array IBLOCK must be at least max(1,N).

On exit: at each row/column j where e(j) is zero or negligible, the matrix T is considered to split
into a block diagonal matrix and IBLOCK(i) contains the block number of the eigenvalue stored in
W(i), for i = 1, 2, . . . , m. Note that IBLOCK(i) < 0 for some i whenever INFO = 1 (see Section 5).

16: ISPLIT(∗) — INTEGER array Global Output

Note: the dimension of the array ISPLIT must be at least max(1,N).

On exit: the leading NSPLIT elements contain the points at which the matrix T splits up into
submatrices (blocks) as follows. The first submatrix consists of rows/columns 1 to ISPLIT(1), the
second submatrix consists of rows/columns ISPLIT(1) + 1 to ISPLIT(2), and the last submatrix
consists of rows/columns ISPLIT(NSPLIT−1) + 1 to ISPLIT(NSPLIT) (= n).

17: WORK(∗) — DOUBLE PRECISION array Local Workspace

Note: the dimension of WORK must be at least max(1,LWORK). The minimum value of LWORK
required to successfully call this routine can be obtained by setting LWORK = −1. The required
size is returned in array element WORK(1).

On exit: WORK(1) contains the minimum dimension of the array WORK required to successfully
complete the task.

18: LWORK — INTEGER Local Input

On entry: either −1 (see WORK) or the dimension of the array WORK required to successfully
complete the task. If LWORK is set to −1 on entry this routine simply performs some initial error
checking and then, if these checks are successful, calculates the minimum size of LWORK required.

Constraints:

either LWORK = −1
or LWORK ≥ max(7,5×N).

19: IWORK(∗) — INTEGER array Local Workspace

Note: the minimum value of LIWORK required to successfully call this routine can be obtained
by setting LIWORK = −1. The required size is returned in array element IWORK(1).

On exit: IWORK(1) contains the minimum dimension of array IWORK required to successfully
complete the task.

20: LIWORK — INTEGER Local Input

On entry: either −1 (see IWORK) or the dimension of the array WORK required to successfully
complete the task. If LIWORK is set to −1 on entry this routine simply performs some initial error
checking and then, if these checks are successful, calculates the minimum size of LIWORK required.

Constraints:

either LIWORK = −1
or LIWORK ≥ max(14,4×N).

21: INFO — INTEGER Global Output

The NAG Parallel Library provides a mechanism, via the routine Z02EAFP, to reduce the amount
of parameter validation performed by this routine. For a full description refer to the Z02 Chapter
Introduction.

On exit: INFO = 0 (or −9999 if reduced error checking is enabled) unless the routine detects an
error (see Section 5).

5 Errors and Warnings

If INFO �= 0 an explanatory message is output and control returned to the calling program.

INFO = −i

On entry, the ith argument was invalid. This error occurred either because a global argument did
not have the same value on all logical processors, or because its value on one or more processors
was incorrect. An explanatory message distinguishes between these two cases.

INFO = 1

The algorithm failed to compute some (or all) of the required eigenvalues to the desired accuracy.
More precisely, IBLOCK(i)< 0 indicates that the ith eigenvalue (stored in W(i)) failed to converge.
The effect is that the eigenvalues may not be as accurate as specified by the tolerance.

INFO = 2

There is a mismatch between the number of eigenvalues computed and the desired number.

INFO = 3

No eigenvalues have been computed. The floating-point arithmetic on the computer is not behaving
as expected.

If failures with INFO ≥ 1 are causing persistent trouble and the user has checked that the routine is
being called correctly, please contact NAG.

6 Further Comments
6.1 Algorithmic Detail

The diagonal matrix D(τ) of the LDLT of the shifted tridiagonal matrix T − τI are computed where τ
is a shift. The inertia of the matrix T − τI is then given by D(τ). The number of negative values of the
diagonal entries of D(τ) gives the number of eigenvalues of T in the interval (−∞, τ ] which is the crucial
information required to locate eigenvalues. See Kahan [1].

6.2 Parallelism Detail

The algorithm is wholly parallel and inter-process communication is minimal.

6.3 Accuracy

See Blackford et al. [2].

8 Example

The example program illustrates the computation of eigenvalues of a 10 by 10 dense symmetric matrix
As. The (i, j)th element of this matrix As is taken to be max(i, j) and the matrix is generated using
the routine F01ZQFP. The value UPLO = ’U’ is used so that only the upper triangular part of As is
referenced.

The matrix As is first tridiagonalised using the routine F08FEFP (PDSYTRD).

The diagonal vector d and the off-diagonal vector e are gathered to each logical processor by calling the
routine F01ZPFP twice.

Finally, the vector d (denoted locally by the array DL) and the the vector e (denoted locally by the array
EL) are used by the routine F08JJFP (PDSTEBZ) to compute the eigenvalues of the tridiagonal T .

8.1 Example Text

* F08JJFP Example Program Text
* NAG Parallel Library Release 3 Revised. NAG Copyright 1999.
* .. Parameters ..

INTEGER NOUT
PARAMETER (NOUT=6)
INTEGER N
PARAMETER (N=12)
INTEGER NA
PARAMETER (NA=20)
INTEGER MP, NP
PARAMETER (MP=2,NP=2)
INTEGER NB
PARAMETER (NB=2)
INTEGER LDA, TDA
PARAMETER (LDA=NA/MP+NB,TDA=NA/NP+NB)
INTEGER LWORK, LIWORK
PARAMETER (LWORK=80,LIWORK=80)

* .. Local Scalars ..
INTEGER I, IA, ICNTXT, IFAIL, INFO, JA, M, NSPLIT,

+ MPROC, NPROC
LOGICAL ROOT
CHARACTER UPLO

* .. Local Arrays ..
DOUBLE PRECISION A(LDA,TDA), D(TDA), DL(N), E(TDA), EL(N),

+ TAU(TDA), W(N), WORK(LWORK)
INTEGER IBLOCK(N), IDESCA(9), ISPLIT(N), IWORK(LIWORK)

* .. External Functions ..
LOGICAL Z01ACFP
EXTERNAL Z01ACFP

* .. External Subroutines ..
EXTERNAL F01ZPFP, F01ZQFP, F08FEFP, F08JJFP, GMATA,

+ Z01AAFP, Z01ABFP
* .. Executable Statements ..

ROOT = Z01ACFP()
IF (ROOT) THEN

WRITE (NOUT,*) ’F08JJFP Example Program Results’
WRITE (NOUT,*)

END IF
*

MPROC = MP
NPROC = NP
IFAIL = 0
CALL Z01AAFP(ICNTXT,MPROC,NPROC,IFAIL)

*
* Set the starting address and array descriptor for A
*

IA = 1
JA = 1
IDESCA(1) = 1
IDESCA(2) = ICNTXT
IDESCA(3) = NA
IDESCA(4) = NA
IDESCA(5) = NB
IDESCA(6) = NB
IDESCA(7) = 0
IDESCA(8) = 0
IDESCA(9) = LDA

*
* Generate the matrix A
*

IFAIL = 0
CALL F01ZQFP(GMATA,N,N,A,IA,JA,IDESCA,IFAIL)

*
* Reduce A to tridiagonal form
*

UPLO = ’U’
CALL F08FEFP(UPLO,N,A,IA,JA,IDESCA,D,E,TAU,WORK,LWORK,INFO)

*
* Gather the diagonal D to each logical processor
*

IFAIL = 0
CALL F01ZPFP(N,IA,JA,IDESCA,D,DL,WORK,LWORK,IFAIL)

*
* Gather the off-diagonal E to each logical processor
*

IFAIL = 0
CALL F01ZPFP(N,IA,JA,IDESCA,E,EL,WORK,LWORK,IFAIL)

*
* Compute the eigenvalues of T (and of A)
*

IFAIL = 0
IF (UPLO.EQ.’L’) THEN

CALL F08JJFP(ICNTXT,’A’,’E’,N,0.0D0,0.0D0,0,0,-1.0D0,DL,EL,M,
+ NSPLIT,W,IBLOCK,ISPLIT,WORK,LWORK,IWORK,LIWORK,
+ INFO)
ELSE IF (UPLO.EQ.’U’) THEN

CALL F08JJFP(ICNTXT,’A’,’E’,N,0.0D0,0.0D0,0,0,-1.0D0,DL,EL(2),
+ M,NSPLIT,W,IBLOCK,ISPLIT,WORK,LWORK,IWORK,LIWORK,
+ INFO)
END IF

*

* Print the computed eigenvalues
*

IF (ROOT) THEN
WRITE (NOUT,*) ’Eigenvalues’
WRITE (NOUT,*)
DO 20 I = 1, N

WRITE (NOUT,*) I, W(I)
20 CONTINUE

END IF
*

IFAIL = 0
CALL Z01ABFP(ICNTXT,’N’,IFAIL)

*
STOP
END

*
SUBROUTINE GMATA(I1,I2,J1,J2,AL,LDAL)

*
* GMATA generates the block A(I1: I2, J1: J2) of the upper
* triangular part of the matrix A such that
*
* a(i,j) = max(i,j)
*
* in the array AL.
*
* .. Scalar Arguments ..

INTEGER I1, I2, J1, J2, LDAL
* .. Array Arguments ..

DOUBLE PRECISION AL(LDAL,*)
* .. Local Scalars ..

INTEGER I, J, K, L
* .. Intrinsic Functions ..

INTRINSIC MAX
* .. Executable Statements ..

L = 1
DO 40 J = J1, J2

K = 1
DO 20 I = I1, I2

IF (I.LE.J) THEN
AL(K,L) = MAX(I,J)

END IF
K = K + 1

20 CONTINUE
L = L + 1

40 CONTINUE
*

RETURN
*

END

8.2 Example Data

None.

[NP3344/3/pdf] F08JJFP (PDSTEBZ).7



F08JJFP (PDSTEBZ) NAG Parallel Library Manual

8.3 Example Results

F08JJFP Example Program Results

Eigenvalues

1 -18.2653445961538
2 -3.91505976820175
3 -1.74107763047893
4 -1.01013587928877
5 -0.678451626799023
6 -0.501676952612396
7 -0.397980864225928
8 -0.333705083138936
9 -0.293061251985502

10 -0.268013496211147
11 -0.254347826770162
12 105.658854975866

